Advanced Waypoint Tutorial

(For Game Maker 4.0 and above)

By Allen Cheung

(Allen_Cheung@hotmail.com)
1 Introduction

First of all, many thanks to Mark Overmars for creating Game Maker (now 4.1!). Also thanks to many other people of the GM community and forums, for continuing to support the program and fostering its reputation.

Game Maker: http://www.cs.uu.nl/~markov/gmaker/index.html
GM Forums: http://pub58.ezboard.com/bgamemakercommunity
So why waypoints? Waypoints are used mostly for Artificial Intelligence (AI), giving an easy way to implement complex motions and checking for various conditions without having some other rigorous system. Using the system is easy; making it, however, takes a bit more work, but is definitely doable in GM. Especially now, with future GM versions having support for paths (which is basically waypoints), it might be worthwhile to see how such a system would work.

Because of the complexity of the tutorial, however, I thought a document explaining how it works would be best. I’m going to go through, chronologically, how I made the tutorial; that is usually how we write programs in the first place, and this way I can point out various issues I had to overcome.

A bit on notation:

Discussion is in normal font.

Instructions are written in bold font.

Terminology, objects, sprites, etc. are underlined.

Actual code is typed in Courier New.

2 Getting Started

Waypoints consist of two things; points and connecting lines. From each point, there can be any number of connecting lines to other points, and objects will move from point to point via those lines. More useless trivia – this system of points and lines, with a specified direction, is called a directed graph in Computer Science. =)

Anyway, I find it easiest to begin a program by making all the parts I need.

Create the following basic resources.

Sprites: points (sprPoint), controller (sprMouse).

Sounds: none

Background: none

Scripts: none

Objects: points (objPoint), controller (objMouseController)

Rooms: main room (roomMain)

I reason I use sprMouse and objMouseController is merely because I plan to use the mouse as the controlling device.

Something that you may miss just looking at the .gmd file is that the important sprites all have their origins placed in the center – i.e. a 16x16 sprite will have its origin at (8, 8). While it is not imperative that you do this, most of the time (especially with mouse-controlled games) this just makes the game look nicer, with your mouse relating to the center of objects. Just run the program to see what I mean. 

Set the origin of sprPoint to the center of the sprite.

Set the sprite of objPoint to sprPoint, objMouseController to sprMouse.
One last thing we’ll try to do is allow the user to add points by left-clicking in the window. We do this with the objMouseController. 

Add an instance of objMouseController to roomMain. In the Step event:

if (mouse_check_button(mb_left)) {
  instance_create(mouse_x, mouse_y, objPoint);
  mouse_clear(mb_left);
}

We use mouse_check_button() to directly check whether the button has been pressed or not (opposed to using the Left Button event). Note also that we clear the mouse after we create the instance; otherwise, dragging the button will create a string of instances. 

Save and test your program.

3 Setting Lines, Snapping

Now, let’s get down to business. We want to be able to drag the right mouse button to connect any two points. Therefore, for each point, we need a list of all the lines stemming out from it. After a bit of thought, though, we realize that this is redundant – we can actually just keep a list of all the next possible points, and we can draw “dummy lines” between points. To do this, every point needs to keep track of all its next points.

In the Create event of objPoint:

// Initialize a bunch of variables
numNextPoints = 0;
for (i = 1; i <= 10; i += 1) {

  arrNextPoint[i] = 0;
  arrNextProb[i] = 0;
}

Technically, we don’t have to initialize any arrays, but I suppose it’s force of habit to “declare” variables before I use them.

When we drag, we want to have some indicator of where we’re dragging from and to. For this, we create a objLine object, which is to be created whenever we click the right mouse button, and destroyed after we release it.

Create a objLine object. In its Drawing event:

// Draw line from (x,y) to mouse coords
pen_color = c_black; pen_size = 2;
draw_line(x,y,mouse_x,mouse_y);
Basically, the line will be redrawn whenever the mouse moves, and will connect from its origin to wherever your mouse is. Observe that because of this, the line’s origin must be where you start drawing the line, i.e. the first point.

Now, we use our friend objMouseController to detect any right mouse clicks. We detect when the button is pressed, and when the button is released after we hold it down. We need to store the coordinates of the two points…then we realize we might as well store the two points (as objects) themselves.

Another concept I would like to add here is snapping. GM already uses such a device – you can snap to a grid and check whether your object is aligned (snapped) onto a grid. If you have used any Macromedia product (Dreamweaver, Flash, etc.), you will know what I mean. Simply, when you get close enough to some object, you automatically snap onto that object. Implementing this actually isn’t that hard, so we’ll try it. We will define a constant, as the distance for snapping, though.

In the Create event of objMouseController:

// Variables for mouse-related stuff
mouseState = 0;
x1 = 0; y1 = 0;
x2 = 0; y2 = 0;
firstPoint = 0; secondPoint = 0;
global.SNAPDIS = 20;
To make snapping work, we need to find the distance between the mouse and the object. We can do this any number of ways, but it sounds like a function that we can use later on, so why not make it into a script?

Create a findDistance script:

// Finds distance between (arg0, arg1) and object arg2 closest to that point
// Used for "snapping"
{
  Obj = instance_nearest(argument0, argument1, argument2);
  dx = argument0 - Obj.x; dy = argument1 - Obj.y;
  return sqrt(dx * dx + dy * dy);
}
We just calculate the distance with the Pythagorean theorem, a^2 + b^2 = c^2. Note that this script returns a value that we can use.

Detecting when the user first clicks the right mouse button by adding this in the Step event of objMouseController:

// Checks right mouse press
// mouseState indicates dragging
// 0 = not pressed, 1 = released
if (mouse_check_button(mb_right) && mouseState == 0) 
  if (findDistance(mouse_x, mouse_y, objPoint) < global.SNAPDIS) {

    instance_create(Obj.x, Obj.y, objLine);
    mouseState = 1;
    x1 = Obj.x; y1 = Obj.y;
    firstPoint = Obj;
  }     

As you can see, we use findDistance() here to our advantage. Also note that we’re slightly “abusing” scripts here. The variable Obj was actually created in the script, but we still have full access to it outside the script here; that’s why accessing it here works. We save all our variables for later, and change mouseState.
Detecting when the user releases the mouse button, more code for objMouseController’s Step event:

if (!mouse_check_button(mb_right) && mouseState == 1) {
  with (objLine) instance_destroy();
  if (findDistance(mouse_x, mouse_y, objPoint) < global.SNAPDIS) {
    x2 = Obj.x; y2 = Obj.y;
    secondPoint = Obj;
  // Mark the two as connected only if they're two separate points
  // Also limit the number of possible paths per point to 10
    if (x1 != x2 && y1 != y2 && firstPoint.numNextPoints <= 10) {
      firstPoint.numNextPoints += 1;  
      firstPoint.arrNextPoint[firstPoint.numNextPoints] = secondPoint;
    }
  }
  // Finally, just call create event to reset everything
  event_perform(ev_create,0);
}

More code here. The first thing we do is destroy the objLine instance – we don’t need it anymore. Then we mirror what we need for the first part, in saving our variables and snapping close to the object. After that, we have to do a bit of processing; we check to see whether the user is dragging back to the same object (which would cancel the line), then increment the number of points in our first point and add the second point to the first point’s list (which would mean that we’re moving from the first point to the second). In effect, you have produced this:

Point 1                                Point 2



arrNextPoint[1] = Point 2                           Point 3

arrNextPoint[2] = Point 3

…

With the connecting lines controlled by the user’s dragging. One more thing – whether we cancel our line or connect it to another object, we need to reset all our variables (firstPoint, x1, etc.). We could do it manually, but we notice that in objMouseController’s own Creation event, we set all those up. Therefore, it’s a simple call to event_perform() to reset all our variables.

Save, and test. At this point, you should be able to lay down points, and drag lines between them. You can’t see those lines visually, however.

4 More Line Drawing

Breathe a sigh of relief – the most complex ideas are behind us. Now that we have the basic engine set up, let’s take care of a few details here and there. For one, it would be nice if we could actually see the lines that we’re drawing/connecting, right?

We could make dozens of objConnectingLine objects, but we immediately run into problems; how would we draw lines of all different angles? We notice that GM has a draw_line() command, but where would we place it? Remember that all our information between points is stored IN points, more precisely, in the point that we’re originating from. So, it would only make sense for us to draw lines there. 

In the Drawing event of objPoint, enter the following code:

// Draw itself, and connections to other points
draw_sprite(sprPoint, 0, x, y);
for (i = 1; i <= numNextPoints; i += 1) {

  otherPoint = arrNextPoint[i];
  dx = otherPoint.x - x; dy = otherPoint.y - y;
  draw_line(x, y, otherPoint.x, otherPoint.y);
  draw_circle(x + dx/5, y + dy/5, 5);
}

Remember to draw the original sprite of the point, since we override it when we use the Drawing event! Next, we just draw a line between every point that is connected to our first point. To make everything neat, I have also drawn a circle a 1/5th of the way from our originating point. A diagram may make sense of this:

                                                  (x2, y2)

     (x1+dx/5,y1+dy/5)                    dy


             dy/5

      dx/5

(x1,y1)              dx

So in effect, we have drawn little circles to indicate the direction that we’re going. 

As usual, save and test. You should be able to see those lines that you’re drawing.

5 Bullet-Proof, Probabilities

If you’ve fooled around with the program up to this point, you might notice that if you right-click with no points, your program crashes. This is because under the Step event of objMouseController, it makes use of objPoint, which does not exist when there are no points on the field. This is obviously not very good…who knows what mischief users can cause? 

The concept of making sure that the user doesn’t send everything crashing is sometimes called bullet-proofing. As the name implies, it means that you add extra code and checks to make sure everything runs smoothly. We will take care of that little problem with a check.

Modify your code under objMouseController’s Step event:

if (instance_number(objPoint) > 0) {

  if (mouse_check_button(mb_right) && mouseState == 0) {

  .. ..

  }

  if (!mouse_check_button(mb_right) && mouseState == 1) {

  .. ..

  }

}

This just checks that there are, in fact, instances of objPoint before you start dragging lines between them. Simple, elegant, and effective.

Now that we have a good base to build from, let’s just add one little fancy thing here. At a point, an object can choose where to go next – how that happens is up to you. I’ve chosen a mechanism where you can input an integer, and the object will randomly choose between lines based on their integers (the higher the better). 

To accomplish this, we need to keep track of every probability number of every line; since we’re keeping information on the next points in our original points anyway, we might as well keep information on their probabilities in there too. As a matter of fact, if you were observant, you would note that we already declared that in the beginning with the variable arrNextProb[], an array that is meant to complement arrNextPoint[]. 

Modify objMouseController’s Step event to the following:

if (x1 != x2 && y1 != y2 && firstPoint.numNextPoints <= 10) {

  firstPoint.numNextPoints += 1;  
  prob = get_integer("Probability of transversing this path? ", 1);
  firstPoint.arrNextPoint[firstPoint.numNextPoints] = secondPoint;
  firstPoint.arrNextProb[firstPoint.numNextPoints] = prob;
}

We ask for the probability, and then store it inside the same loop that we store the point.

One more fanciful thing. We can print the probabilities of every line on the screen; this is done relatively similar to the way we drew circles 1/5th of the way across the line. 

Modify the Drawing event in the objPoint object:

// Draw itself, and connections to other points with prob. number
draw_sprite(sprPoint, 0, x, y);
for (i = 1; i <= numNextPoints; i += 1) {

  otherPoint = arrNextPoint[i];
  dx = otherPoint.x - x; dy = otherPoint.y - y;
  draw_line(x, y, otherPoint.x, otherPoint.y);
  draw_circle(x + dx/5, y + dy/5, 5);
  font_color = c_black; font_style = fs_normal; font_size = 12;
  draw_text(x + dx/2, y + dy/2 - 20, arrNextProb[i]);
}

As always, save and test. 

6 Moving on the Track

Now comes the fun part – watching your little objects run around your custom point-line-point combinations. I’ll just call the entire structure a track from now on, though. 

First, we need something for us to run around with.

Create some sprite that you’d like to see slide around; the tutorial uses sprGhost. Make sure to set the origin to the sprite’s center.

Create an object from that sprite; the tutorial makes objGhost.

We need a way to move those ghosts as well. Since we’ve already covered dragging, let’s try another (and perhaps simpler) technique – click once to move, click again to drop. However, it only makes sense for us to “pick up” ghosts when we click on them, so we go to sprGhost’s Left Button event. For this, we need a variable to keep check of whether we’re moving the ghost or just moving the mouse. 

While we’re at it, though, let’s think ahead. What else do we need? We need something to keep track of whether the ghost is just sitting around, or it’s actually locked onto a track and ready to rumble. It would also be nice if we knew where we were heading – the next point. (note how this pre-production thinking is already suggesting a way to code the actual movement)

Put this in the Creation event of sprGhost:

// Init. variables 
// whether ghost is "marked" (moving via mouse) 
marked = false;
// 0 = not locked onto track, 1 = locked at point, 2 = traveling on line
locked = 0;
// Reference to next destination point
nextPoint = 0;
// Variable for ghost speed
SPEED = 10;
I also gave ghosts a constant speed, although that is probably unnecessary (as all objects have an internal speed variable anyway). Just thought I’d illustrate putting a constant in a noticeable place.

In the Left Button event of sprGhost:

// If unmarked, then mark it
if (!marked) {

  marked = true;
  locked = 0;
  nextPoint = 0;
  mouse_clear(mb_left);
}

else {

  if (findDistance(mouse_x, mouse_y, objPoint) < global.SNAPDIS) {

    x = Obj.x; y = Obj.y;
    locked = 1;
    nextPoint = Obj;
  }

  marked = false;
  mouse_clear(mb_left);
}

The code should be self-explanatory. If the object was previously unmarked (being picked up), then we mark it, and reset all its variables (because we can pick up the ghost on the track as well).; as a precaution, we clear the mouse to make sure nothing bad happens if we hold onto the left button too long. On the other hand, if the object was marked (being put down), we use a few pieces of familiar code – we snap it, save the variables, and change the locked status (it is now locked onto a track). And regardless of whether we can snap it onto a point, we unmark the object and clear the mouse.

Save and run. Try picking up your objects.

Hm…we run into a strange problem. Remember objMouseController? It reads in directly from the mouse, which means that it’ll plop down a point even when you’re moving objects around. It’s not destructive, but is pretty annoying and can look sloppy. To cure us of this ailment, we just make a check in objMouseController.

Modify the code in the Step event of objMouseController to look like the following:

// Checks left mouse press
if (mouse_check_button(mb_left)) {

  // If moving ghosts, then don't place new objects
  with (objGhost)
    if (marked) exit;
  instance_create(mouse_x, mouse_y, objPoint);
  mouse_clear(mb_left);

}

We just check to see if it’s marked, and if so, ignore the code to place points.

…and the final piece of code to get this working!

Place this under the Step event of sprGhost:

// If marked, just follow the cursor
if (marked) {

  x = mouse_x;
  y = mouse_y;
} else 
// If locked onto track, start moving
// Stop at the "end of the line"
if (locked == 1) {

  if (nextPoint.numNextPoints > 0) {

    // Random prob.
    sum = 0;
    for (i = 1; i <= nextPoint.numNextPoints; i += 1) 
      sum += nextPoint.arrNextProb[i];
    ranNum = floor(random(sum)+1);
    // Determine next point
    sum = 0; i = 0;
    while (sum < ranNum) {

      i += 1;
      sum += nextPoint.arrNextProb[i];
    }

    nextPoint = nextPoint.arrNextPoint[i];
    // start moving
    locked = 2;
    move_towards_point(nextPoint.x, nextPoint.y, SPEED);
  }

} else 
// Moving along track, careful to check for point
if (locked == 2) { 
  dx = nextPoint.x - x; dy = nextPoint.y - y;
  distance = sqrt(dx*dx + dy*dy);
  if (distance <= SPEED/2) {

    x = nextPoint.x; y = nextPoint.y;
    speed = 0;
    locked = 1;
  }

}

It’s not as intimidating as it looks, of course. First off, we write code to move the object along with the mouse cursor if we’re moving it. Then, if we’re detecting that the object is locked onto the track AND that there are still more points to go (it’s not the end of the track, so to speak), we do a fancy probability algorithm. In this case, a picture indeed equals a thousand words:


                          2


                          1


                 1



Probs =                                     2                                    1                        1

The first loop adds everything up; sum = 4.

We then generate a random number between 1 and sum.

The second loop then keeps on adding probabilities until it reaches that number. In our example, the random number was 3, and the loop had to add 2 and 1 to get there. The variable i, therefore, is at 2.

We set our destination (nextPoint), then begin to move there. 

Finally, when locked = 2, we’re moving on the track, and the one thing that we need to check for is when we reach our destination. In this case, we check for whether the object is close enough – it is when it reaches SPEED/2. Why this number? Another diagram:




                             speed

The rectangles represent our object. The way that GM moves is by “jumping” pixels every step; in this case, it jumps SPEED pixels every time we call it to move. In many cases, though, it will jump past the point and continue going on…we can’t simply check for whether its coordinates are the same as the point’s. As you can see, however, when we check the distance, we know that the object will come within SPEED/2 units of the point, either before it or after. Therefore, we choose this number to check for proximity.

Note that we can’t use findDistance() there, because we already know our destination and our coordinates. Actually…now that I look at it, we could have used the function point_distance() to find that distance rather than calculating it outright (these things happen when you review your own code…), but since our code works, why change it? An engineer’s adage: “If it ain’t broke, don’t fix it.”

Save and run. If everything works as planned (by both you and me), then you should have your objects running around a track in no time. Note that you can create linear tracks, circular tracks, figure-eights, whatever…the engine only cares about individual points, and really doesn’t know about the entire track.

7 Conclusion

Well, that should have been somewhat entertaining; it certainly was for me. I have added a reset button to the tutorial to make things easier, but feel free to play around with various parameters, algorithms, and whatnot…this isn’t exactly a game, but tweaking it will help you understand and ultimately use such a system/engine in your own games. Then again, there’s a certain appeal to watching objects shift around in a systematic fashion…

As always, I’m aware that there are many potential mistakes here. The tutorial works fine, but this document was written after the tutorial was wholly completed, so I might have skipped a few things on my quest for a chronological layout. If you spot any such embarrassing mistakes, please don’t hesitate to e-mail me.

Anyhoo, there should be a few more tutorials/guides coming; I personally still have a few concepts and ideas that could be useful to the rest of the GM community. So look for those, and please give me feedback on the ones that currently exist! 

“A successful tool is one that was used to do something undreamed of by its author.”

· S. C. Johnson (very true for GM!)

